

Access

Teil 3

Tabellenverarbeitung in VBA

Begleitheft



Inhalt

Lektion 1	Datensätze auswählen - Datenbank AT01.accdb	4
Lektion 2	Datensatzauswahl im Recordset – Datenbank AT01.accdb	5
Lektion 3	Datensätze löschen – Datenbank AT01.accdb	6
Lektion 4	Datensätze einfügen – Datenbank AT01.accdb	7
Lektion 5	Datensätze suchen – Datenbank AT01.accdb	9
Lektion 6	Datensätze bearbeiten – Datenbank AT01.accdb	11
Lektion 7	Verwendung von Datums- und Zeitangaben – Datenbank AT02.accdb	12
Lektion 8	Wochenplan (Anwendungsbeispiel) – Datenbank AT02.accdb	13
Lektion 9	Datensatzquelle im Formular – Datenbank AT02.accdb	14
Lektion 10	Tabellenverarbeitung mit gebundenen Feldern– Datenbank AT02.accdb	15
Lektion 11	Datensätze im Formular bearbeiten – Datenbank AT02.accdb	16
Lektion 12	Datensätze im Formular generieren – Datenbank AT02.accdb	19
Lektion 13	Tabellenverarbeitung sperren – Datenbank AT02.accdb	20
Lektion 14	Formular mit temporärer Tabelle – Datenbank AT03.accdb	22
Lektion 15	Temporäre Tabelle in VBA generieren – Datenbank AT03.accdb	23
Lektion 16	Tabellenverknüpfung in VBA erstellen – Datenbank AT04.accdb	24

Einführung


Mit dem Videobuch „Access – Tabellenverarbeitung in VBA“ wird die Einführung in Access um die Tabellenverarbeitung in VBA erweitert.

Zunächst wird auf die unterschiedlichen Abfragetypen und die Deklaration eines Recordsets über eine SQL-Anweisung eingegangen.

Das Buch enthält eine Einführung in SQL, wobei bei der Formulierung der SQL-Anweisungen auf die SQL-Ansicht von Abfragen zugegriffen wird, um einen schnellen Einstieg zu erhalten.

Es werden unterschiedliche Zugriffsmethoden erläutert, um innerhalb von VBA Tabellensätze anzufügen, zu löschen, zu editieren oder Datensätze innerhalb von Recordsets zu suchen.

Ferner wird die Verwendung von VBA-Code innerhalb von Formularen gezeigt, um auf Recordsets in unterschiedlichsten Formen zugreifen zu können. Es werden Funktionalitäten eines Formulars dargestellt, die für eine praxisgerechte Nutzung unumgänglich sind, sich jedoch nur mittels VBA lösen lassen.



Beispiel zur Personaleinsatzplanung

Eine wichtige Besonderheit ist die dynamische Zuordnung der Datensatzquelle eines Formulars während der Bearbeitung in Abhängigkeit von den Eingaben des Benutzers sowie die Verwendung einer zentralen Parametertabelle, um die Funktionalität der Anwendung an die Vorgaben des Benutzers flexibel ausrichten zu können. Hierzu wird ein komplexes Beispiel zur Personaleinsatzplanung mit graphischen Elementen vorgestellt.

Zum Abschluss wird gezeigt, wie bei einer Mehrplatzanwendung automatisch die Tabellen-Verweise von Front- und Backend einer Anwendung mittels VBA verwaltet werden können, sogar dann, wenn der Endanwender nur das Runtime-System im Einsatz hat. Dabei wird auch auf den File-Dialog eingegangen.

Lektion 1 Datensätze auswählen - Datenbank AT01.accdb

tblBuch						
BuchID	Verfasser	Titel	Auflage	Erscheinungsjahr	Erscheinungsort	VerlagID
1	Arnold, Dieter	Handbuch Logistik		2008	Berlin	42
2	Blum, H. S.	Logistik-Controlling - Kor		2006	Wiesbaden	35
3	Brühl, Rolf	Controlling – Grundlager		2012	München	39
4	Corbat, Peter	Logistik in Vertriebsunte		2009	Norderstedt	32
5	Delfman/Reihlen	Controlling von Logistikp		2003	Stuttgart	41
6	Domschke, Wolfgang	Logistik: Rundreisen und		1997	München	39
7	Ehrmann, Harald	Logistik		2003	Ludwigshafen	38
8	Eschenbach, Rolf	Controlling		1995	Stuttgart	41
9	Fiedler, Rolf	Einführung in das Contro		1998	München	39
10	Galsmayer, Johannes	Performance Measureme		2004	Düsseldorf	44
11	Göpfert, Ingrid	Logistik – Führungskonze		2005	München	45
12	Gudehuss, Tim	Logistik	4. Auflage	2010	Berlin	42
13	Walter, Hans-Christian	Systementwicklung	5. Auflage	2007	Köln	47
14	Weber, Jürgen	Logistik- und Supply Cha		2002	Stuttgart	41
15	Werner, Hartmut	Supply Chain Manageme		2010	Wiesbaden	36
16	Zorn, Richard	Musterthema		2015	Berlin	
(Neu)						0

tblVerlag			
VerlagID	Verlag	Ort	Land
32	Books on Demand	Norderstedt	D
33	by-mi Fachverlag	Landsberg /Lech	D
34	Carl Hanser Verlag	München	D
35	Deutscher Universitätsverlag	Wiesbaden	D
36	Gabler Verlag	Wiesbaden	D
37	GRIN Verlag	Norderstedt	D
38	Kiehl Friedrich Verlag	Ludwigshafen	D
39	Oldenbourg Verlag	München	D
40	Physica Verlag	Würzburg	D
41	Schäffer-Poeschel Verlag	Stuttgart	D
42	Springer Verlag	Berlin	D
43	VDE Verlag	Offenbach	D
44	VDM Verlag Müller	Düsseldorf	D
45	Verlag Franz Vahlen	München	D
46	Verlag Moderne Industrie	Landsberg/Lech	D
47	TÜV Rheinland	Köln	D
48	Vieweg und Teubner Verlag	Wiesbaden	D
49	Musterverlag Bayern	München	D

Verknüpfungseigenschaften

Linker Tabellenname: **tblBuch** Rechter Tabellenname: **tblVerlag**

Linker Spaltenname: **VerlagID** Rechter Spaltenname: **VerlagID**

☒ 1: Beinhaltet nur die Datensätze, bei denen die Inhalte der verknüpften Felder beider Tabellen gleich sind.
☐ 2: Beinhaltet ALLE Datensätze aus 'tblBuch' und nur die Datensätze aus 'tblVerlag', bei denen die Inhalte der verknüpften Felder beider Tabellen gleich sind.
☐ 3: Beinhaltet ALLE Datensätze aus 'tblVerlag' und nur die Datensätze aus 'tblBuch', bei denen die Inhalte der verknüpften Felder beider Tabellen gleich sind.

OK Abbrechen Neu

qryBuch01

tblBuch

- * BuchID
- Verfasser
- Titel
- Auflage
- Erscheinungsjahr
- Erscheinungsort
- VerlagID

tblVerlag

- * VerlagID
- Verlag
- Ort
- Land

Feld: **tblBuch.*** BuchID Erscheinungsjahr

Tabelle: **tblBuch** tblBuch tblBuch

Sortierung: Aufsteigend

Anzeigen: ☒ ☐ ☐ ☐ ☐

Kriterien: >=2010

oder:

```


SELECT tblBuch.*, tblVerlag.*
FROM tblBuch INNER JOIN tblVerlag ON tblBuch.VerlagID = tblVerlag.VerlagID
WHERE (((tblBuch.Erscheinungsjahr)>=2010))
ORDER BY tblBuch.BuchID;

```

SQL-Ansicht der Abfrage

Lektion 2 Datensatzauswahl im Recordset – Datenbank AT01.accdb

Abfrage



BuchID	Verfasser	Titel	Erscheint	Erscheinur	tblBuch.Verlag	tblBuch.VerlagID	tblVerlag.VerlagID	tblVerlag.Verlag	Ort	Land
12	Gudehuss, Tim	Logistik	2010	Berlin	Spriger Verlag		105	105 Spriger Verlag		
16	Zorn, Richard	Musterther	2015	Berlin						

```
SELECT tblBuch.*, tblVerlag.*
FROM tblBuch LEFT JOIN tblVerlag ON tblBuch.VerlagID = tblVerlag.VerlagID
WHERE (((tblBuch.Erscheinungsort)="Berlin") AND ((tblBuch.Erscheinungsjahr)>=2010))
ORDER BY tblBuch.BuchID;
```

SQL-Ansicht der Abfrage

```
' Modul: modAT01_Satzauswahl
```

```
Option Compare Database
Option Explicit
```

```
Sub SatzAuswahl()
Dim db As Database
Dim rs As Recordset
Dim sSql As String
Set db = CurrentDb
```

```
'Verwendung der Abfrage
```

```
Set rs = db.OpenRecordset("qryBuch_Verlag01")
```

```
'alternativ, Verwendung der SQL-Anweisung
```

```
sSql = " SELECT tblBuch.*, tblVerlag.* " & _
" From tblBuch Left Join tblVerlag On tblBuch.VerlagID = tblVerlag.VerlagID " & _
" Where Erscheinungsjahr >= 2010 " & _
" And Erscheinungsort = 'Berlin' " & _
" Order By BuchID"
```

```
Set rs = db.OpenRecordset(sSql)
```

```
If rs.RecordCount > 0 Then
```

```
rs.MoveFirst
```

```
Do Until rs.EOF
```

```
MsgBox rs!Verfasser & ": " & rs!Titel & " | " & rs!Erscheinungsjahr & _
```

```
" - " & rs!Verlag
```

```
rs.MoveNext
```

```
Loop
```


```
End If
```

```
MsgBox "Ende"
```



```
End Sub
```

Datenbank		
Deklaration	Dim db As Database	Variable vom Typ Datenbank
Zuweisung	Set db = CurrentDb	Aktuelle Datenbank zuweisen
Methoden		
	db.OpenRecordset (Tabelle, Abfrage, SQL-Anweisung)	Öffnen
	db.Execute (SQL-Anweisung)	Ausführung
Recordset		
Deklaration	Dim rs As Recordset	Variable vom Typ Recordset
Zuweisung	Set rs = db.OpenRecordset (...)	Datenquelle zuweisen
Methoden		
	rs.MoveFirst / rs.MoveNext	Gehe zum ersten / nächsten Datensatz
	rs.AddNew	Gehe zu einem neuen Datensatz
	rs.Edit	Bearbeitungsmodus für Datensatz herstellen
	rs.Update	Datensatz in die Datenbank zurückschreiben
	rs.Delete	Datensatz löschen
	rs.FindFirst / rs.FindNext (Kriterium)	Suche Datensatz
Eigenschaften		
	rs.RecordCount	Anzahl Datensätze des Recordsets
	rs.EOF	True, wenn kein Datensatz beim Move mehr vorliegt
	rs.NoMatch	True, wenn kein Datensatz bei Find gefunden wird
Felder		
	rs!Feldname	Feld eines Datensatzes

Lektion 3 Datensätze löschen – Datenbank AT01.accdb



Löschen-Abfrage



SQL-Ansicht der Abfrage

```
' Modul: modAT02_Löschen 1
```

```
Option Compare Database
Option Explicit
```

```
Sub Löschen()
Dim db As Database
Set db = CurrentDb
'Löschen Sätze in einer Tabelle mittels Abfrage
db.Execute "qryBuch_Löschen01"
```

```
'Löschen Sätze in Tabelle Buch mittels SQL-Anweisung
Dim sSql As String
sSql = "Delete * from tblBuch"
db.Execute sSql
```


Übernahme in den VBA-Code

```
'Löschen Sätze in Tabelle Buch mittels Einzelanweisung
Dim rsBuch As Recordset
Set rsBuch = db.OpenRecordset("tblBuch")
If rsBuch.RecordCount > 0 Then
    rsBuch.MoveFirst
    Do Until rsBuch.EOF
        rsBuch.Delete
        rsBuch.MoveNext
    Loop
End If
End Sub
```

```
' Modul: modAT02_Löschen 2
```

```
Sub Löschen()
Dim db As Database
Set db = CurrentDb
Dim sSql As String
'Löschen Sätze in Tabelle Buch mittels Einzelanweisung
Dim rsBuch As Recordset
sSql = "Select * from tblBuch where Titel Like '*Logistik*'"
Set rsBuch = db.OpenRecordset(sSql)
If rsBuch.RecordCount > 0 Then
    rsBuch.MoveFirst
    Do Until rsBuch.EOF
        rsBuch.Delete
        rsBuch.MoveNext
    Loop
End If
End Sub
```

Lektion 4 Datensätze einfügen – Datenbank AT01.accdb



```
' Modul: modAT03_Hinzufügen
```

```
Option Compare Database
Option Explicit
```

```
Sub Hinzufügen()
Dim db As Database
Set db = CurrentDb
```

```
'Anfügen Datensätze mittels Abfrage
db.Execute "qryBuchImport_AnfügenBuch01"
```

```
'Anfügen Datensätze mittels SQL-Anweisung
```

```
Dim sSql As String
sSql = "INSERT INTO tblBuch ( Verfasser, Titel, Auflage, Erscheinungsjahr, Erscheinungsort, Verlag )" & _
" SELECT Verfasser, Titel, Auflage, Erscheinungsjahr, Erscheinungsort, Verlag FROM tblBuchImport" & _
" WHERE Erscheinungsjahr >=2000 AND Erscheinungsort = 'München'"
db.Execute sSql
```

Übernahme in den VBA-Code

Abfrage1

Auswahl-Abfrage

tblBuchImport

BuchID
Verfasser
Titel
Auflage
Erscheinungsjahr
Erscheinungsort
Verlag

Feld:	tblBuchImport.*	Erscheinungsort	Erscheinungsjahr			
Tabelle:	tblBuchImport	tblBuchImport	tblBuchImport			
Sortierung:		Aufsteigend	Absteigend			
Anzeigen:	☒	☒	☒	☐	☐	☐
Kriterien:		"München"				
oder:			>=2010			

Abfrage1

SELECT tblBuchImport.*, tblBuchImport.Erscheinungsort, tblBuchImport.Erscheinungsjahr
FROM tblBuchImport
WHERE (((tblBuchImport.Erscheinungsort)="München")) OR (((tblBuchImport.Erscheinungsjahr)>=2010))
ORDER BY tblBuchImport.Erscheinungsort, tblBuchImport.Erscheinungsjahr DESC;

SQL-Ansicht der Abfrage

'Anfügen Datensätze in Tabelle Buch mittels Einzelanweisung
Dim rsBuchImport As Recordset
Dim rsBuch As Recordset

```
sSql = "Select * from tblBuchImport" & _
      " Where Erscheinungsjahr >= 2010 OR Erscheinungsort = 'München'" & _
      " Order By Erscheinungsort, Erscheinungsjahr desc"
Set rsBuchImport = db.OpenRecordset(sSql)
Set rsBuch = db.OpenRecordset("tblBuch")
If rsBuchImport.RecordCount > 0 Then
    rsBuchImport.MoveFirst
    Do Until rsBuchImport.EOF
        rsBuch.AddNew
        rsBuch!Verfasser = rsBuchImport!Verfasser
        rsBuch!Titel = rsBuchImport!Titel
        rsBuch!Auflage = rsBuchImport!Auflage
        rsBuch!Erscheinungsjahr = rsBuchImport!Erscheinungsjahr
        rsBuch!Erscheinungsort = rsBuchImport!Erscheinungsort
        rsBuch.Update
        rsBuchImport.MoveNext
    Loop
End If
End Sub
```

Übernahme in den VBA-Code

Lektion 5 Datensätze suchen – Datenbank AT01.accdb

Datensätze für Erscheinungsjahr im Recordset suchen

BuchID	Verfasser	Titel	Erscheinungsjahr	Erscheinungsort
1	Arnold, Dieter	Handbuch Logistik	2008	Berlin
2	Blum, H. S.	Logistik-Controlling - Kontext, A	2006	Wiesbaden
3	Brühl, Rolf	Controlling – Grundlagen des Er	2012	München
4	Corbat, Peter	Logistik in Vertriebsunternehme	2009	Norderstedt
5	Delfman/Reihlen	Controlling von Logistikprozesse	2003	Stuttgart
6	Domschke, Wolfgang	Logistik: Rundreisen und Touren	1997	München
7	Ehrmann, Harald	Logistik	2003	Ludwigshafen
8	Eschenbach, Rolf	Controlling	1995	Stuttgart
9	Fiedler, Rolf	Einführung in das Controlling	1998	München
10	Gaismayer, Johannes	Performance Measurement für	2004	Düsseldorf
11	Göpfert, Ingrid	Logistik – Führungskonzeption	2005	München
12	Gudehuss, Tim	Logistik	2010	Berlin
13	Walter, Hans-Christian	Systementwicklung	2007	Köln
14	Weber, Jürgen	Logistik- und Supply Chain Cont	2002	Stuttgart
15	Werner, Hartmut	Supply Chain Management: Gru	2010	Wiesbaden
16	Zorn, Richard	Musterthema	2015	Berlin

```

' Modul: modAT05_DatensätzeSuchen 1

```

```

Option Compare Database
Option Explicit

```

```

Sub DatensätzeSuchen()
Dim db As Database
Dim rs As Recordset
Dim sKriterium As String
Dim sAntwort As String

```

```

Set db = CurrentDb
Set rs = db.OpenRecordset("Select * from tblBuch")
If rs.RecordCount = 0 Then Exit Sub
Do
    sAntwort = InputBox("Bitte geben Sie das Erscheinungsjahr ein.")
    If sAntwort = "" Then Exit Do
    If IsNumeric(sAntwort) Then
        sKriterium = "Erscheinungsjahr = " & sAntwort
        rs.FindFirst (sKriterium)
        Do Until rs.NoMatch
            MsgBox rs!BuchID & vbCrLf & _
                rs!Verfasser & vbCrLf & _
                rs!Titel & vbCrLf & _
                rs!Erscheinungsjahr & vbCrLf & _
                rs!Erscheinungsort
            rs.FindNext (sKriterium)
        Loop
    End If
    MsgBox "Ende der Suche!"
Loop
End Sub

```

Datensätze für Erscheinungsort im Recordset suchen

BuchID	Verfasser	Titel	Erscheinungsjahr	Erscheinungsort
1	Arnold, Dieter	Handbuch Logistik	2008	Berlin
2	Blum, H. S.	Logistik-Controlling - Kontext, A	2006	Wiesbaden
3	Brühl, Rolf	Controlling – Grundlagen des Er	2012	München
4	Corbat, Peter	Logistik in Vertriebsunternehme	2009	Norderstedt
5	Delfman/Reihlen	Controlling von Logistikprozesse	2003	Stuttgart
6	Domschke, Wolfgang	Logistik: Rundreisen und Tourer	1997	München
7	Ehrmann, Harald	Logistik	2003	Ludwigshafen
8	Eschenbach, Rolf	Controlling	1995	Stuttgart
9	Fiedler, Rolf	Einführung in das Controlling	1998	München
10	Gaismayer, Johannes	Performance Measurement für I	2004	Düsseldorf
11	Göpfert, Ingrid	Logistik – Führungskonzeption	2005	München
12	Gudehuss, Tim	Logistik	2010	Berlin
13	Walter, Hans-Christian	Systementwicklung	2007	Köln
14	Weber, Jürgen	Logistik- und Supply Chain Cont	2002	Stuttgart
15	Werner, Hartmut	Supply Chain Management: Gru	2010	Wiesbaden
16	Zorn, Richard	Musterthema	2015	Berlin

```
' Modul: modAT05_DatensätzeSuchen 2
```

```
Option Compare Database
Option Explicit
```

```
Sub DatensätzeSuchen2()
Dim db As Database
Dim rs As Recordset
Dim sKriterium As String
Dim sAntwort As String
```

```
Set db = CurrentDb
Set rs = db.OpenRecordset("Select * tblBuch")
If rs.RecordCount = 0 Then Exit Sub
Do
    sAntwort = InputBox("Bitte geben Sie den Erscheinungsort ein.", vbYesNo)
    If sAntwort = "" Then Exit Do
    sKriterium = "Erscheinungsort = '" & sAntwort & "'"
    rs.FindFirst (sKriterium)
    Do Until rs.NoMatch
        MsgBox rs!BuchID & vbCrLf & _
            rs!Verfasser & vbCrLf & _
            rs!Titel & vbCrLf & _
            rs!Erscheinungsjahr & vbCrLf & _
            rs!Erscheinungsort
        rs.FindNext (sKriterium)
    Loop
    MsgBox "Ende der Suche!"
Loop
End Sub
```

Lektion 6 Datensätze bearbeiten – Datenbank AT01.accdb

tblBuch						
BuchID	Verfasser	Titel	Erscheinungsjahr	Ort	Verlag	VerlagID
1	Arnold, Dieter	Handbuch L	2008	Berlin	Spriger Verlag	
2	Blum, H. S.	Logistik-Cor	2006	Wiesbaden	Deutscher Universitätsverlag	
3	Brühl, Rolf	Controlling	2012	München	Oldenbourg Verlag	
4	Corbat, Peter	Logistik in V	2009	Norderstedt	Books on Demand	
5	Delfman/Reihlen	Controlling	2003	Stuttgart	Schäffer-Poeschel Verlag	
6	Domschke, Wolfga	Logistik: Ru	1997	München	Oldenbourg Verlag	
7	Ehrmann, Harald	Logistik	2003	Ludwigshafen	Kiehl Friedrich Verlag	
8	Eschenbach, Rolf	Controlling	1995	Stuttgart	Schäffer-Poeschel Verlag	
9	Fiedler, Rolf	Einführung	1998	München	Oldenbourg Verlag	
10	Gaismayer, Johann	Performanc	2004	Düsseldorf	VDM Verlag Müller	
11	Göpfert, Ingrid	Logistik – Fi	2005	München	Verlag Franz Vahlen	
12	Gudehuss, Tim	Logistik	2010	Berlin	Spriger Verlag	
13	Walter, Hans-Chris	Systementv	2007	Köln	Verlag TÜV Rheinland	
14	Weber, Jürgen	Logistik- un	2002	Stuttgart	Schäffer-Poeschel Verlag	
15	Werner, Hartmut	Supply Chai	2010	Wiesbaden	Gabler Verlag	
16	Zorn, Richard	Musterthen	2015	Berlin		

tblVerlag				
VerlagID	Verlag	Ort	Land	Zum Hinzufügen klicken
105	Spriger Verlag			
*(Neu)				

```

' Modul: modAT06_VerlagIDErgänzen

```

```

Option Compare Database
Option Explicit

```

```

Sub VerlagIDErgänzen()
Dim db As Database
Dim rsBuch As Recordset
Dim rsVerlag As Recordset
Dim sKriterium As String

```

```

Set db = CurrentDb
Set rsBuch = db.OpenRecordset("tblBuch")
Set rsVerlag = db.OpenRecordset("Select * from tblVerlag")

```

```

If rsBuch.RecordCount > 0 Then
    rsBuch.MoveFirst
    Do Until rsBuch.EOF
        If rsBuch!Verlag <> "" Then
            rsBuch.Edit
            sKriterium = "Verlag = " & rsBuch!Verlag & ""
            rsVerlag.FindFirst (sKriterium)
            If rsVerlag.NoMatch Then
                rsVerlag.AddNew
                rsVerlag!Verlag = rsBuch!Verlag
                rsBuch!VerlagID = rsVerlag!VerlagID 'Übertragung Primärschlüssel
                rsVerlag.Update
            Else
                rsBuch!VerlagID = rsVerlag!VerlagID 'Übertragung Primärschlüssel
            End If
            rsBuch.Update
        End If
        rsBuch.MoveNext
    Loop
End If

```

```

MsgBox "Ende"
End Sub

```

Lektion 7 Verwendung von Datums- und Zeitangaben – Datenbank AT02.accdb

```

'-----
' Modul: modDemoDate
'-----
Option Compare Database
Option Explicit
Sub DemoDate()
Dim date1 As Date
Dim d1 As Double
'1.
date1 = "30.12.1899 00:00:00": d1 = date1 'Nulldatum
Debug.Print "1. Datum: " & Format(date1, "d.m.yyyy hh:mm:ss") & " ... in Double: " & d1
'2.
'1 Tag = 24 * 60 Min = 1440 Min | 2 Std 24 Min = 144 Min = 144 1/1440 Tage = 0,1 Tage
date1 = "30.12.1899 02:24:00": d1 = date1
Debug.Print "2. Datum: " & Format(date1, "d.m.yyyy hh:mm:ss") & " ... in Double: " & d1
'3.
date1 = "31.12.1899 00:00:00": d1 = date1
Debug.Print "3. Datum: " & Format(date1, "d.m.yyyy hh:mm:ss") & " ... in Double: " & d1
'4.
date1 = "31.12.1899 12:00:00": d1 = date1
Debug.Print "4. Datum: " & Format(date1, "d.m.yyyy hh:mm:ss") & " ... in Double: " & d1
'5.
date1 = "31.12.1900 06:00:00": d1 = date1
Debug.Print "5. Datum: " & Format(date1, "d.m.yyyy hh:mm:ss") & " ... in Double: " & d1
'6.
' (42003 Tage - 28 Tage aus Schaltjahren) / 365 = 115 Jahre | 1899 + 115 = 2014
date1 = "30.12.2014 00:00:00": d1 = date1
Debug.Print "6. Datum: " & Format(date1, "d.m.yyyy hh:mm:ss") & " ... in Double: " & d1
'7. - 13.
date1 = d1 + 20 + 5 / 24: d1 = date1
Debug.Print "7. Datum: " & Format(date1, "dd.mm.yyyy hh:mm:ss") & " ... in Double: " & d1
Debug.Print "8. Datum: " & Format(date1, "hhmm") & " ... in Double: " & d1
Debug.Print "9. Datum: " & Format(date1, "m/d/yyyy") & " ... in Double: " & d1
Debug.Print "10. Datum: " & Format(date1, "mVdVyyyy") & " ... in Double: " & d1
Debug.Print "11. Datum: " & Format(date1, "dddd \der dd.mmmm yyyy") & " ... in Double: " & d1
Debug.Print "12. Datum: " & Format(date1, "hh U\hr mm") & " ... in Double: " & d1
'13.
date1 = date1 + 1 / (24 * 60): d1 = date1
Debug.Print "13. Zeit: " & Format(date1, "dd.mm.yyyy hh:mm") & " ... in Double: " & d1
'14
date1 = "05:10": d1 = date1
Debug.Print "14. Datum: " & Format(date1, "dd.mm.yyyy hh:mm:ss") & " ... in Double: " & d1
End Sub

```

Direktbereich

```

1. Datum: 30.12.1899 00:00:00 ... in Double: 0
2. Datum: 30.12.1899 02:24:00 ... in Double: 0,1
3. Datum: 31.12.1899 00:00:00 ... in Double: 1
4. Datum: 31.12.1899 12:00:00 ... in Double: 1,5
5. Datum: 31.12.1900 06:00:00 ... in Double: 366,25
6. Datum: 30.12.2014 00:00:00 ... in Double: 42003
7. Datum: 19.01.2015 05:00:00 ... in Double: 42023,2083333333
8. Datum: 0500 ... in Double: 42023,2083333333
9. Datum: 1.19.2015 ... in Double: 42023,2083333333
10. Datum: 1/19/2015 ... in Double: 42023,2083333333
11. Datum: Montag der 19.Januar 2015 ... in Double: 42023,2083333333
12. Datum: 05 Uhr 00 ... in Double: 42023,2083333333
13. Zeit: 19.01.2015 05:01 ... in Double: 42023,2090277778
14. Datum: 30.12.1899 05:10:00 ... in Double: 0,215277777777778

```

Lektion 8 Wochenplan (Anwendungsbeispiel) – Datenbank AT02.accdb

Formular

Access Tabellenverarbeitung in VBA | Datensatzquelle im Formular

Wochenplan

Datum	Wochentag	Mitarbeiter	Schicht	Von	Bis	7	8	9	10	11	12	13	14	15	16	17	18	Arbeitszeit	Soll
01.06.2015	Mo	Karmelk, Anna	1	06:45	11:45		06:45											5,00	5,00
01.06.2015	Mo	Müller, Rolf	1	06:45	10:45													4,00	4,00
01.06.2015	Mo	Büttner, Marianne	2	09:00	14:00													5,00	5,00
01.06.2015	Mo	Wegener, Klaus	2	09:00	13:00													4,00	4,00
01.06.2015	Mo	Krause, Hans	3	11:00	18:30													7,50	7,50
01.06.2015	Mo	Schlüter, Rita	3	11:00	18:30													7,50	7,50
02.06.2015	Di	Karmelk, Anna	1	06:45	11:45													5,00	5,00
02.06.2015	Di	Müller, Rolf	1	06:45	10:45													4,00	4,00
02.06.2015	Di	Büttner, Marianne	2	09:00	14:00													5,00	5,00
02.06.2015	Di	Wegener, Klaus	2	09:00	13:00													4,00	4,00
02.06.2015	Di	Krause, Hans	3	11:00	18:30													7,50	7,50
02.06.2015	Di	Schlüter, Rita	3	11:00	18:30													7,50	7,50
03.06.2015	Mi	Karmelk, Anna	1	06:45	11:45													5,00	5,00
03.06.2015	Mi	Müller, Rolf	1	06:45	10:45													4,00	4,00
03.06.2015	Mi	Büttner, Marianne	2	09:00	14:00													5,00	5,00
03.06.2015	Mi	Wegener, Klaus	2	09:00	13:00													4,00	4,00
03.06.2015	Mi	Krause, Hans	3	11:00	18:30													7,50	7,50
03.06.2015	Mi	Schlüter, Rita	3	11:00	18:30													7,50	7,50
04.06.2015	Do	Karmelk, Anna	1	06:45	11:45													5,00	5,00
04.06.2015	Do	Müller, Rolf	1	06:45	10:45													4,00	4,00
04.06.2015	Do	Büttner, Marianne	2	09:00	14:00													5,00	5,00
04.06.2015	Do	Wegener, Klaus	2	09:00	13:00													4,00	4,00
04.06.2015	Do	Krause, Hans	3	11:00	18:30													7,50	7,50
04.06.2015	Do	Schlüter, Rita	3	11:00	18:30													7,50	7,50
05.06.2015	Fr	Karmelk, Anna	1	06:45	11:45													5,00	5,00
05.06.2015	Fr	Müller, Rolf	1	06:45	10:45													4,00	4,00
05.06.2015	Fr	Büttner, Marianne	2	09:00	14:00													5,00	5,00
05.06.2015	Fr	Wegener, Klaus	2	09:00	13:00													4,00	4,00
05.06.2015	Fr	Krause, Hans	3	11:00	18:30													7,50	7,50
05.06.2015	Fr	Schlüter, Rita	3	11:00	18:30													7,50	7,50
06.06.2015	Mo	Karmelk, Anna	1	06:45	11:45													5,00	5,00

Ab Datum: 01.06.2015 Anzahl Tage: 100 Zeitraum angrenzen Mitarbeiter auswählen Plan generieren Gesamt: 495 495

Datensatz: 1 von 90 Ungefiltert Suchen

Access Tabellenverarbeitung in VBA - Demo

TABELLENTOOLS FELDER TABELLE

Alle Access... Suchen...

Tabellen

- tblMitarbeiter Tabelle
Erstellt am: 16.02.201...
Geändert am: 21.02.2...
- tblParameter Tabelle
Erstellt am: 06.02.201...
Geändert am: 21.02.2...
- tblPlan Tabelle
Erstellt am: 16.02.201...
Geändert am: 21.02.2...

Abfragen

- qryPlan Abfrage
Erstellt am: 16.02.201...
Geändert am: 11.03.2...

Formulare

- frmPlan Formular
Erstellt am: 06.02.201...
Geändert am: 04.03.2...
- frmPlan02 Formular
Erstellt am: 10.03.201...
Geändert am: 11.03.2...

Module

tblMitarbeiter

MitarbeiterID	Mitarbeiter	Arbeitszeit	Schicht	Bemerkung	Zum Hinzufügen klicken
1	Krause, Hans	7,5	3		
2	Schlüter, Rita	7,5	3		
3	Büttner, Marianne	5	2		
4	Karmelk, Anna	5	1		
5	Müller, Rolf	4	1		
6	Wegener, Klaus	4	2		

Datensatz: 1 von 6 Kein Filter Suchen

tblParameter

ParameterID	Parameter	Bezeichnung	Systemparameter	Zum Hinzufügen klicken
10		Balken Voll	☒	
11		Balken Leer	☒	
12	7 8	Balken Überschrift	☒	
20	6:45	Tag Start, z.B. 06:45	☐	
21	18:30	Tag Ende, z.B. 17:30	☐	
22	06:45	Schicht 1 (Frühschicht), z.B. 06:45	☐	
23	09:00	Schicht 2 (Mittelschicht), z.B. 09:00	☐	
24	11:00	Schicht 3 (Spätschicht), z.B. 10:00	☐	

Datensatz: 1 von 8 Kein Filter Suchen

tblPlan

ID	Datum	Wochentag	Mitarbeiter	Von	Bis		
1316	23.02.2015		Müller, Rolf	06:45	10:45		
1311	23.02.2015		Karmelk, Anna	06:45	11:45		
1296	23.02.2015		Krause, Hans	11:00	18:30		
1306	23.02.2015		Büttner, Marianne	09:00	14:00		
1301	23.02.2015		Schlüter, Rita	11:00	18:30		
1321	23.02.2015		Wegener, Klaus	09:00	13:00		
1302	24.02.2015		Schlüter, Rita	11:00	18:30		
1312	24.02.2015		Karmelk, Anna	06:45	11:45		
1322	24.02.2015		Wegener, Klaus	09:00	13:00		
1307	24.02.2015		Büttner, Marianne	09:00	14:00		
1297	24.02.2015		Krause, Hans	11:00	18:30		
1317	24.02.2015		Müller, Rolf	06:45	10:45		

Datensatz: 1 von 180 Kein Filter Suchen

Lektion 9 Datensatzquelle im Formular – Datenbank AT02.accdb

Tabellen

tblPlan		
Feldname	Felddatentyp	Beschreibung (optional)
ID	AutoWert	
Datum	Datum/Uhrzeit	
MitarbeiterID	Zahl	
Schicht	Zahl	
Von	Datum/Uhrzeit	
Bis	Datum/Uhrzeit	
Balken	Kurzer Text	

tblMitarbeiter		
Feldname	Felddatentyp	Beschreibung (optional)
MitarbeiterID	AutoWert	
Mitarbeiter	Kurzer Text	
Arbeitszeit	Zahl	
Schicht	Zahl	
Bemerkung	Kurzer Text	

Abfrage

qryPlan

Feld:	ID	Datum	Tag: Format([Datum];"tttt")	MitarbeiterID	Schicht	Von	Bis	Balken	Arbeitszeit: Runden(([Bis]-[Von])*24;2)	Soll: Arbeitszeit	Mitarbeiter
Tabelle:	tblPlan	tblPlan		tblPlan	tblPlan	tblPlan	tblPlan	tblPlan		tblMitarbeiter	tblMitarbeiter
Sortierung:		Aufsteigend				Aufsteigend					Aufsteigend
Anzeigen:	☒	☒	☒	☒	☒	☒	☒	☒	☒	☒	☒
Kriterien:											
oder:											

Datensatzquelle als Eigenschaft im Formular festlegen

frmPlan

Formularkopf

Datum	Wochentag	Mitarbeiter	Schicht	Von	Bis	Ungebunden
-------	-----------	-------------	---------	-----	-----	------------

Detailbereich

Datum	Tag	MitarbeiterID	Schicht	Von	Bis
-------	-----	---------------	---------	-----	-----

Formularfuß

Ab Datum Ungebunden Anzahl Tage gebunden Zeitraum angrenzen Mitarbeiter auswählen Ungebunden

Eigenschaftenblatt

Auswahltyp: Formular

Formular

Format	Daten	Ereignis	Andere	Alle
Datensatzquelle	qryPlan			
Recordsettyp	Dynaset			
Standardwerte abrufen	Ja			
Filter				
Beim Laden filtern	Nein			
Sortiert nach	[qryPlan].[Datum]			
Beim Laden sortieren	Ja			
Auf Nachverarbeitung warten	Nein			
Daten eingeben	Nein			
Anfügen zulassen	Ja			
Löschen zulassen	Ja			
Bearbeitungen zulassen	Ja			
Filter zulassen	Ja			
Datensätze sperren	Keine Sperrungen			

Lektion 10 Tabellenverarbeitung mit gebundenen Feldern– Datenbank AT02.accdb

Abfrage

The screenshot shows the 'qryPlan' query design view. It features two tables: 'tblPlan' and 'tblMitarbeiter'. 'tblPlan' contains fields: ID (primary key), Datum, Wochentag, MitarbeiterID, Von, Bis, Balken, and Schicht. 'tblMitarbeiter' contains fields: MitarbeiterID (primary key), Mitarbeiter, Arbeitszeit, Schicht, and Bemerkung. A relationship line connects the 'MitarbeiterID' field in 'tblPlan' to the 'MitarbeiterID' field in 'tblMitarbeiter'. Below the design view, the 'Abfragesetup' (Query Setup) grid is displayed, showing the fields and their properties for the query.

Feld:	ID	Datum	Tag: Format([Datum], "ttt")	MitarbeiterID	Schicht	Von	Bis	Balken	Arbeitszeit: Runden([
Tabelle:	tblPlan	tblPlan		tblPlan	tblPlan	tblPlan	tblPlan	tblPlan	
Sortierung:		Aufsteigend				Aufsteigend			
Anzeigen:	☒	☒	☒	☒	☒	☒	☒	☒	☒
Kriterien:									
oder:									

Formularentwurf

The screenshot shows the 'frmPlan' form design view. It features a form with various controls. The 'Formularkopf' (Form Header) section contains fields for 'Datum', 'Wochentag', 'Mitarbeiter', 'Schicht', 'Von', 'Bis', 'Ungebunden', 'Arbeitszeit', and 'Soll'. The 'Detailbereich' (Detail Section) contains fields for 'Datum', 'Tag', 'MitarbeiterID', 'Schicht', 'Von', 'Bis', 'Balken', and 'Arbeitszeit: Soll'. The 'Formularfuß' (Form Footer) section contains a summary row with fields for 'Ab Datum', 'Ungebunden', 'Anzahl Tage', 'Gebunden', 'Zeitraum abgrenzen', 'Mitarbeiter auswählen', 'Ungebunden', 'Plan generieren', and 'Gesamt: =Summe([=Summe('

Feldeigenschaften

Objekttyp	Name	Feldinhalt	Aktiviert	Gesperrt	Ereignisprozedur
Feld	Datum	Gebunden an Feld Datum der Datensatzquelle	Ja	Nein	
Feld	Tag	Gebunden an Feld Tag der Datensatzquelle	Ja	Ja	
Feld	MitarbeiterID	Gebunden an Feld MitarbeiterID der Datensatzquelle	Ja	Nein	
Feld	Schicht	Gebunden an Feld Schicht der Datensatzquelle	Ja	Nein	Nach Aktualisierung
Feld	Von	Gebunden an Feld Von der Datensatzquelle	Ja	Nein	Nach Aktualisierung
Feld	Bis	Gebunden an Feld Bis der Datensatzquelle	Ja	Nein	Nach Aktualisierung
Feld	Balken	Gebunden an Feld Balken der Datensatzquelle	Nein	Ja	
Feld	Arbeitszeit	Gebunden an Feld Arbeitszeit der Datensatzquelle	Nein	Ja	
Feld	Soll	Gebunden an Feld Soll der Datensatzquelle	Nein	Ja	
Feld	SummeArbeitszeit	= Summe(Arbeitszeit) Funktion	Nein	Ja	
Feld	SummeSoll	= Summe(Soll) Funktion	Nein	Ja	
Feld	AbDatum	Ungebunden	Ja	Nein	
Feld	BalkenÜberschrift	Ungebunden	Ja	Nein	Nach Aktualisierung
Feld	TagStart	Ungebunden	Nein	Ja	
Feld	TagEnde	Ungebunden	Nein	Ja	
Feld	AnzTage	Ungebunden	Ja	Nein	
Feld	MitarbeiterAnzeigen	Ungebunden	Ja	Nein	Nach Aktualisierung

Lektion 11 Datensätze im Formular bearbeiten – Datenbank AT02.accdb

```
'-----
' Formularmodul: Form_frmPlan
'-----
```

```
Option Compare Database
Option Explicit
```

```
'-----
' Variablen auf Modulebene
'-----
```

```
Dim sVoll As String      'Zeichen für Balken: Anwesend
Dim sLeer As String     'Zeichen für Balken: Abwesend
Dim sÜberschrift As String 'Überschriftszeile für Balken
```

```
Dim dateTagVon As Date   'Zeitpunkt Tagesbeginn in Balkendarstellung
Dim dateTagBis As Date  'Zeitpunkt Tagesende in Balkendarstellung
Dim datSchicht1 As Date 'Zeitpunkt Beginn Frühschicht
Dim datSchicht2 As Date 'Zeitpunkt Beginn Mittelschicht
Dim datSchicht3 As Date 'Zeitpunkt Beginn Spätschicht
```

```
'-----
' Ereignisprozeduren
'-----
```

```
Private Sub Form_Open(Cancel As Integer) 'Beim Öffnen des Formulars
ParameterSetzen
Me.TagStart = dateTagVon
Me.TagEnde = dateTagBis
DatensatzquelleAlle 'Tageseinträge aller Mitarbeiter im abgegrenzen Zeitraum
End Sub
```

```
Private Sub BalkenÜberschrift_AfterUpdate() 'Nach Aktualisierung Balkenüberschrift wird Parameter neu gesetzt
Dim db As Database
Set db = CurrentDb
Dim rsParameter As Recordset
Set rsParameter = db.OpenRecordset("Select * from tblParameter")
rsParameter.FindFirst "ParameterID = " & 12
rsParameter.Edit
rsParameter!Parameter = Me.BalkenÜberschrift
rsParameter.Update
End Sub
```

```
Private Sub Schicht_AfterUpdate() 'Nach Aktualisierung Schicht
If IsNull(Me.Schicht) Then Me.Schicht = 0
Me.Von = DateZeitVon(Me.Schicht)
Von_AfterUpdate
End Sub
```

```
Private Sub Von_AfterUpdate() 'Nach Aktualisierung Von (Beginnzeitpunkt)
Me.Bis = Me.Von + Me.Soll / 24
Me.Balken = ""
If Not IsNull(Me.Von) And Not IsNull(Me.Bis) Then Me.Balken = sBalken(Me.Von, Me.Bis)
End Sub
```

```
Private Sub Bis_AfterUpdate() 'Nach Aktualisierung Bis (Endzeitpunkt)
Me.Balken = ""
If Not IsNull(Me.Von) And Not IsNull(Me.Bis) Then Me.Balken = sBalken(Me.Von, Me.Bis)
End Sub
```

```
Private Sub MitarbeiterAnzeigen_AfterUpdate() 'Nach Aktualisierung MitarbeiterAnzeigen (Kombinationsfeld)
DatenquelleMitarbeiter
End Sub
```

```
Private Sub AbDatum_DbClick(Cancel As Integer) 'Bei Doppelklick Abgrenzungsdatum
Me.AbDatum = Date 'Standardwert = aktuelles Datum
End Sub
```

```
Private Sub AnzTage_DbClick(Cancel As Integer) 'Bei Doppelklick auf Anzahl Tage für Abgrenzung
Me.AnzTage = 5 'Standardwert = 5 Tage
End Sub
```

```
Private Sub ZeitraumAbgrenzen_Click() 'Bei Klick Schaltfläche Zeitraum abgrenzen
If Me.MitarbeiterAuswählen.Caption = "Mitarbeiter auswählen" Then
DatenatzquelleAlle 'Tageseinträge aller Mitarbeiter im abgegrenzen Zeitraum
Else
DatenquelleMitarbeiter 'Tageseinträge einzelner Mitarbeiter im abgegrenzen Zeitraum
End If
End Sub
```

```

Private Sub MitarbeiterAuswählen_Click() 'Bei Klick Schaltfläche Mitarbeiter Auswählen
    If Me.MitarbeiterAuswählen.Caption = "Mitarbeiter auswählen" Then
        Me.MitarbeiterAuswählen.Caption = "Alle Mitarbeiter"
        Me.MitarbeiterAnzeigen.Visible = True
        Me.MitarbeiterAnzeigen.SetFocus
        Me.MitarbeiterAnzeigen.DropDown 'Kombinationsfeld öffnen
    Else
        Me.MitarbeiterAuswählen.Caption = "Mitarbeiter auswählen"
        Me.MitarbeiterAnzeigen.Visible = False
        DatensatzquelleAlle 'Tageseinträge aller Mitarbeiter im abgegrenzen Zeitraum
    End If
End Sub

```

```

Private Sub PlanGenerieren_Click() 'Bei Klick Schaltfläche Plan generieren
    Dim iAntwort As Long
    iAntwort = MsgBox("Plan wird generiert, möchten Sie fortfahren", vbYesNo)
    If iAntwort = vbNo Then Exit Sub
    Dim db As Database
    Set db = CurrentDb
    Dim rsMitarbeiter As Recordset
    Dim rsPlan As Recordset
    Dim iTag As Long
    Dim dateBisDatum As Date
    Dim sSql As String

    If Me.MitarbeiterAnzeigen.Visible = True Then
        sSql = "Select * from tblMitarbeiter where MitarbeiterID = " & Me.MitarbeiterAnzeigen
    Else
        sSql = "tblMitarbeiter"
    End If

    Set rsMitarbeiter = db.OpenRecordset(sSql)

    If rsMitarbeiter.RecordCount = 0 Then Exit Sub

    Set rsPlan = db.OpenRecordset("tblPlan")
    rsMitarbeiter.MoveFirst

    Do Until rsMitarbeiter.EOF
        For iTag = 0 To Me.AnzTage - 1
            rsPlan.AddNew
            rsPlan!MitarbeiterID = rsMitarbeiter!MitarbeiterID
            rsPlan!Schicht = rsMitarbeiter!Schicht
            If IsNull(rsPlan!Schicht) Then rsPlan!Von = dateTagVon Else rsPlan!Von = DateZeitVon(rsPlan!Schicht)
            rsPlan!Bis = rsPlan!Von + rsMitarbeiter!Arbeitszeit / 24
            rsPlan!Balken = sBalken(rsPlan!Von, rsPlan!Bis)
            rsPlan!Datum = Me.AbDatum + iTag
            rsPlan.Update
        Next iTag
        rsMitarbeiter.MoveNext
    Loop

    dateBisDatum = Me.AbDatum + Me.AnzTage - 1
    sSql = "Select * from qryPlan where Datum >= #" & Format(Me.AbDatum, "m/d/yyyy") & "# And Datum <= #" & Format(dateBisDatum, "m/d/yyyy") & "#
    Me.RecordSource = sSql
End Sub

```

Anwendungsprozeduren

```

Sub ParameterSetzen() 'Parameter setzen
    Dim db As Database
    Set db = CurrentDb
    Dim rsParameter As Recordset
    Set rsParameter = db.OpenRecordset("Select* from tblParameter")

    rsParameter.FindFirst "ParameterID = " & 10
    If IsNull(rsParameter!Parameter) Then sVoll = "." Else sVoll = rsParameter!Parameter

    rsParameter.FindFirst "ParameterID = " & 11
    If IsNull(rsParameter!Parameter) Then sLeer = " " Else sLeer = rsParameter!Parameter

    rsParameter.FindFirst "ParameterID = " & 12
    Me.BalkenÜberschrift = rsParameter!Parameter

    rsParameter.FindFirst "ParameterID = " & 20
    If IsDate(rsParameter!Parameter) Then dateTagVon = Format(rsParameter!Parameter, "hh:mm") Else dateTagVon = "06:00"

```

```

rsParameter.FindFirst "ParameterID = " & 21
If IsDate(rsParameter!Parameter) Then dateTagBis = Format(rsParameter!Parameter, "hh:mm") Else dateTagBis = "18:00"

rsParameter.FindFirst "ParameterID = " & 22
If IsDate(rsParameter!Parameter) Then datSchicht1 = Format(rsParameter!Parameter, "hh:mm")

rsParameter.FindFirst "ParameterID = " & 23
If IsDate(rsParameter!Parameter) Then datSchicht2 = Format(rsParameter!Parameter, "hh:mm")

rsParameter.FindFirst "ParameterID = " & 24
If IsDate(rsParameter!Parameter) Then datSchicht3 = Format(rsParameter!Parameter, "hh:mm")

End Sub

```

```

Sub DatensatzquelleAlle() 'Datensatzquelle setzen: Alle Mitarbeiter im abgegrenzten Zeitraum
Dim dateBisDatum As Date
Dim sSql As String
If IsNull(Me.AbDatum) Then Me.AbDatum = Date
If IsNull(Me.AnzTage) Then Me.AnzTage = 7
dateBisDatum = Me.AbDatum + Me.AnzTage - 1
sSql = "Select * from qryPlan" & _
    " where Datum >= #" & Format(Me.AbDatum, "m\vd\yyyy") & "#" & _
    " And Datum <= #" & Format(dateBisDatum, "m\vd\yyyy") & "#" & _
Me.RecordSource = sSql
End Sub

```

```

Sub DatenquelleMitarbeiter() 'Datensatzquelle setzen: Einzelner Mitarbeiter im abgegrenzten Zeitraum
Dim dateBisDatum As Date
Dim sSql As String
If IsNull(Me.AbDatum) Then Me.AbDatum = Date
If IsNull(Me.AnzTage) Then Me.AnzTage = 7
dateBisDatum = Me.AbDatum + Me.AnzTage - 1
sSql = "Select * from qryPlan" & _
    " where Datum >= #" & Format(Me.AbDatum, "m\vd\yyyy") & "#" & _
    " And Datum <= #" & Format(dateBisDatum, "m\vd\yyyy") & "#" & _
    " and MitarbeiterID = " & Me.MitarbeiterAnzeigen
Me.RecordSource = sSql
End Sub

```

```

Function DateZeitVon(Schicht As Long) As Date 'Umwandlung Schicht in Startzeitpunkt
Select Case Schicht
Case 1: DateZeitVon = Format(datSchicht1, "hh:mm")
Case 2: DateZeitVon = Format(datSchicht2, "hh:mm")
Case 3: DateZeitVon = Format(datSchicht3, "hh:mm")
End Select
End Function

```

```

Function sBalken(dateVon As Date, dateBis As Date) As String 'Balken erstellen für Eintrag
Dim dateZeit As Date
sBalken = ""
For dateZeit = dateTagVon To dateTagBis - 0.000001 Step "00:05:00"
    If Minute(dateZeit) = 0 Then sBalken = sBalken & "|"
    If dateZeit + 0.000001 > dateVon And dateZeit + 0.000001 < dateBis Then sBalken = sBalken & sVoll Else sBalken = sBalken & sLeer
Next dateZeit
End Function

```

Lektion 12 Datensätze im Formular generieren – Datenbank AT02.accdb

```
Private Sub PlanGenerieren_Click() 'Bei Klick Schaltfläche Plan generieren
Dim iAntwort As Long
iAntwort = MsgBox("Plan wird generiert, möchten Sie fortfahren", vbYesNo)
If iAntwort = vbNo Then Exit Sub
Dim db As Database
Set db = CurrentDb
Dim rsMitarbeiter As Recordset
Dim rsPlan As Recordset
Dim iTag As Long
Dim dateBisDatum As Date
Dim sSql As String

If Me.MitarbeiterAnzeigen.Visible = True Then
    sSql = "Select * from tblMitarbeiter where MitarbeiterID = " & Me.MitarbeiterAnzeigen
Else
    sSql = "tblMitarbeiter"
End If

Set rsMitarbeiter = db.OpenRecordset(sSql)

If rsMitarbeiter.RecordCount = 0 Then Exit Sub

Set rsPlan = db.OpenRecordset("tblPlan")
rsMitarbeiter.MoveFirst

Do Until rsMitarbeiter.EOF
    For iTag = 0 To Me.AnzTage - 1
        rsPlan.AddNew
        rsPlan!MitarbeiterID = rsMitarbeiter!MitarbeiterID
        rsPlan!Schicht = rsMitarbeiter!Schicht
        If IsNull(rsPlan!Schicht) Then rsPlan!Von = dateTagVon Else rsPlan!Von = DateZeitVon(rsPlan!Schicht)
        rsPlan!Bis = rsPlan!Von + rsMitarbeiter!Arbeitszeit / 24
        rsPlan!Balken = sBalken(rsPlan!Von, rsPlan!Bis)
        rsPlan!Datum = Me.AbDatum + iTag
        rsPlan.Update
    Next iTag
    rsMitarbeiter.MoveNext
Loop

dateBisDatum = Me.AbDatum + Me.AnzTage - 1
sSql = "Select * from qryPlan where Datum >= #" & Format(Me.AbDatum, "mVd\yyyy") & "# And Datum <= #" & Format(dateBisDatum,
"mVd\yyyy") & "#
Me.RecordSource = sSql
End Sub
```

Lektion 13 Tabellenverarbeitung sperren – Datenbank AT02.accdb

<

```
'-----  
' Modul: Form_frmPlan02  
'-----
```

```
Private Sub Bearbeiten_Click()      'Bei Klick in Bearbeitungsmodus wechseln  
Me.AllowAdditions = True  
Me.AllowDeletions = True
```

```
Me.Datum.Locked = False  
Me.MitarbeiterID.Locked = False  
Me.Schicht.Locked = False  
Me.Von.Locked = False  
Me.Bis.Locked = False
```

```
Me.PlanGenerieren.Visible = True  
Me.Bearbeiten.Caption = "Bearbeitungsmodus"
```

```
Me.Datum.SetFocus  
End Sub
```

Temporäre Tabelle

[illegible]

Abfrage

The screenshot displays the 'qryPlanTemp' query design view. It features two tables: 'tblPlanTemp' and 'tblMitarbeiter'. 'tblPlanTemp' includes fields: ID (primary key), Jahr, Monat, MitarbeiterID, Schicht1, Schicht2, and Schicht3. 'tblMitarbeiter' includes fields: MitarbeiterID (primary key), Mitarbeiter, Arbeitszeit, Schicht, and Bemerkung. A relationship line connects the primary keys of both tables. Below the design view, a table grid is partially visible, showing columns for 'Field:', 'Tabelle:', 'Sortierung:', 'Anzeigen:', and 'Kriterien:'. The rows correspond to the fields: 'MonatTxt: Monatsname', 'Jahr', 'tblPlanTemp', and 'Mitarbeiter'.

Formatierungsregel bearbeiten

Regeltyp auswählen:

- Werte im aktuellen Datensatz prüfen oder einen Ausdruck verwenden
- Mit anderen Datensätzen vergleichen

Regelbeschreibung bearbeiten:

Nur Zellen formatieren, für die gilt:

Feldwert ist

Vorschau: AaBbCcYyZz

OK Abbrechen

Formularentwurf

[illegible]

Formular Plan Übersicht

DATEI **START** **ERSTELLEN** **EXTERNE DATEN** **DATENBANKTOOLS**

Microsoft-Konto

Access Tabellenverarbeitung in VBA - Demo

?

Alle Access-O...

Suchen...

Geändert am: 21.02.201...

tblParameter Tabelle

Erstellt am: 06.02.2015 O...

Geändert am: 21.02.201...

tblPlan Tabelle

Erstellt am: 16.02.2015 1...

Geändert am: 13.03.201...

tblPlanTemp Tabelle

Erstellt am: 13.03.2015 O...

Geändert am: 13.03.201...

Abfragen R

qryPlan Abfrage

Erstellt am: 16.02.2015 O...

Geändert am: 16.02.201...

qryPlanTemp Abfrage

Erstellt am: 13.03.2015 O...

Geändert am: 17.03.201...

Formulare A

frmPlan Formular

Erstellt am: 06.02.2015 O...

Geändert am: 11.03.201...

frmPlan02 Formular

Erstellt am: 10.03.2015 1...

Geändert am: 11.03.201...

frmPlan03 Formular

Erstellt am: 13.03.2015 1...

Geändert am: 16.03.201...

frmPlanÜbers... Formular

Erstellt am: 13.03.2015 O...

Geändert am: 13.03.201...

Module R

Wocheplan

Datum	Wochentag	Mitarbeiter	Schicht	Von	Bis	7	8	9	10	11	12	13	14	15	16	17	18	Arbeitszeit	Soll
04.01.2016	Mo	Karmelk, Anna	1	06:45	11:45													5	5
04.01.2016	Mo	Müller, Rolf	1	06:45	10:45													4	4
04.01.2016	Mo	Buttner, Marianne	2	09:00	14:00													5	5
04.01.2016	Mo	Wegener, Klaus	2	09:00	13:00													4	4
04.01.2016	Mo	Krause, Hans	3	11:00	18:30													7,5	7,5
04.01.2016	Mo	Schlüter, Rita	3	11:00	18:30													7,5	7,5
05.01.2016	Di	Karmelk, Anna	1	06:45	11:45													5	5
05.01.2016	Di	Müller, Rolf	1	06:45	10:45													4	4
05.01.2016	Di	Buttner, Marianne	2	09:00	14:00													5	5
05.01.2016	Di	Wegener, Klaus	2	09:00	13:00													4	4
05.01.2016	Di	Krause, Hans	3	11:00	18:30													7,5	7,5
05.01.2016	Di	Schlüter, Rita	3	11:00	18:30													7,5	7,5
06.01.2016	Mi	Karmelk, Anna	1	06:45	11:45													5	5
06.01.2016	Mi	Müller, Rolf	1	06:45	10:45														
06.01.2016	Mi	Buttner, Marianne	2	09:00	14:00														
06.01.2016	Mi	Wegener, Klaus	2	09:00	13:00														
06.01.2016	Mi	Krause, Hans	3	11:00	18:30														
06.01.2016	Mi	Schlüter, Rita	3	11:00	18:30														
07.01.2016	Do	Karmelk, Anna	1	06:45	11:45														
07.01.2016	Do	Müller, Rolf	1	06:45	10:45														
07.01.2016	Do	Buttner, Marianne	2	09:00	14:00														
07.01.2016	Do	Wegener, Klaus	2	09:00	13:00														
07.01.2016	Do	Krause, Hans	3	11:00	18:30														
07.01.2016	Do	Schlüter																	

Lektion 15 Temporäre Tabelle in VBA generieren – Datenbank AT03.accdb

```

'-----
' Formularmodul: Form_frmPlan03
'-----

```

```

Private Sub PlanÜbersicht_Click() 'Übersicht anzeigen
Save 'Aktuellen Datensatz ggf. speichern
Dim db As Database
Set db = CurrentDb
Dim rs As Recordset 'Datensätze, die aktuell im Formular angezeigt werden
Dim rsTemp As Recordset 'Datensätze der Übersicht
Dim sSuchen As String 'Suchstring

Set rs = Me.RecordsetClone ' 'Datensätze, die aktuell im Formular angezeigt werden
If rs.RecordCount = 0 Then Exit Sub

db.Execute "Delete * from tblPlanTemp" 'Alle Datensätze der Übersicht löschen
Set rsTemp = db.OpenRecordset("Select * from tblPlanTemp") 'Methode FindFirst soll angewendet werden

rs.MoveFirst
Do Until rs.EOF
sSuchen = "MitarbeiterID = " & rs!MitarbeiterID & " And Jahr = " & Year(rs!Datum) & " And Monat = " & Month(rs!Datum)
rsTemp.FindFirst (sSuchen)
If rsTemp.NoMatch Then 'Prüfen, ob Satz schon angelegt wurde, wenn nein ...
rsTemp.AddNew
rsTemp!Jahr = Year(rs!Datum)
rsTemp!Monat = Month(rs!Datum)
rsTemp!MitarbeiterID = rs!MitarbeiterID
Else '... sonst
rsTemp.Edit
End If


rsTemp.Fields("Schicht" & Day(rs!Datum)).Value = rs!Schicht 'Schicht dem Feld Schicht01, ... zuweisen

rsTemp.Update
rs.MoveNext
Loop
DoCmd.Close acForm, "frmPlanÜbersicht" 'Formular Übersicht schließen
DoCmd.OpenForm "frmPlanÜbersicht" 'Formular Übersicht öffnen
End Sub

Sub Save() 'Aktuellen Datensatz speichern
On Error Resume Next
DoCmd.RunCommand acCmdSaveRecord
End Sub

```

Lektion 16 Tabellenverknüpfung in VBA erstellen – Datenbank AT04.accdb



```
'
' Modulmodul: Form_frmPlan03
'
```

```
Option Compare Database
Option Explicit
```

```
Private Sub Form_Open(Cancel As Integer)
    If ppbServerVerknüpfung("tblMitarbeiter") = False Then DoCmd.Quit "Prüfung /Herstellung Verknüpfung zum Server"
End Sub
```

```
Private Sub Personalplan_Click()
    DoCmd.OpenForm "frmPlan03"
End Sub
```

```
Private Sub Befehl1_Click()
    DoCmd.OpenForm "frmMitarbeiter", acFormDS
End Sub
```

```
Private Sub Befehl2_Click()
    DoCmd.OpenForm "frmParameter", acFormDS
End Sub
```

```
'
' Modul: mod_ppbServerVerknüpfung
'
```

```
Option Compare Database
Option Explicit
```

```
' Funktion prüft / aktualisiert Verknüpfungen
'
```

```
' Voraussetzung: Verweis Microsoft Office 15.0 Object Library
```

```
Public Function ppbServerVerknüpfung(sTabelle As String) As Boolean
If bPrüfenDateiverknüpfung(sTabelle) Then
    ppbServerVerknüpfung = True
    Exit Function
Else
    If bVerknüpfungEingabe Then
        If bPrüfenDateiverknüpfung(sTabelle) Then
            MsgBox "Verbindung mit Server wurde hergestellt!", , "H i n w e i s"
            ppbServerVerknüpfung = True
            Exit Function
        End If
    End If
End If
MsgBox "Verknüpfung zu Server kann nicht hergestellt werden, die Anwendung wird geschlossen"
ppbServerVerknüpfung = False
End Function
```

```
Function bPrüfenDateiverknüpfung(sTabelle As String) As Boolean
Dim db As Database
Dim rs As Recordset
On Error GoTo Fehler
Set db = CurrentDb
Set rs = db.OpenRecordset(sTabelle)
bPrüfenDateiverknüpfung = True
Exit Function
Fehler: bPrüfenDateiverknüpfung = False
End Function
```

```
Function bVerknüpfungEingabe() As Boolean
Dim db As Database
Dim tb As TableDef
Dim sServerName As String
sServerName = sServerNameDialog
If sServerName = "" Then
    bVerknüpfungEingabe = False
    Exit Function
End If

Set db = CurrentDb
On Error Resume Next
For Each tb In db.TableDefs
    tb.Connect = ";DATABASE=" & sServerName    'Verknüpfung festlegen
    tb.RefreshLink                          'Verknüpfung aktualisieren
Next tb
bVerknüpfungEingabe = True
End Function
```

```
Function sServerNameDialog() As String
Dim oFileDialog As Office.FileDialog
On Error GoTo Ende
sServerNameDialog = ""
MsgBox "Wählen Sie die Serverdatenbank im nachfolgenden Dialog aus ...", , "H i n w e i s"
Set oFileDialog = Application.FileDialog(msoFileDialogFilePicker)
oFileDialog.Filters.Add "Serverdatenbank auswählen ...", "*.mdb; *.accdb"
Call oFileDialog.Show
sServerNameDialog = oFileDialog.SelectedItems(1)
Ende: Set oFileDialog = Nothing
End Function
```